


# lab 1

## Part 1 :

## Voltage divider



# AC

```
simple_rc_ac.cir | simple_rc_ac.cir

* Circuit Description
* RC low-pass filter with parametric sweep for CPAR

* Parameters
.PARAM CPAR=500p

* Signal sources
V1 1 0 AC 1

* Circuit elements
R1 1 2 1k
C1 2 0 {CPAR}

* Analysis request
* Run ac sweep from 1Hz to 100MEG with 10 pts per decade
.AC DEC 10 1 100MEG
* Use parametric sweep for CPAR: 500p:500p:1.5n
.STEP PARAM CPAR 500p 1.5n 500p
*.STEP PARAM CPAR LIST 500p 1n 1.5n

* Output request
.PRINT AC V(1) V(2)
.PLOT AC V(1) V(2)

* Measure the peak
.MEAS AC PEAK max mag(V(2))

* Measure bandwidth using PEAK/sqrt(2)
.MEAS AC BW WHEN mag(V(2))=PEAK/SQRT(2)

.END
```

LTspice 24.1.9 for Windows

Circuit: C:\Users\DELL\Desktop\Siemens AMS 2025\lab1\simple\_rc\_ac.cir

Start Time: Fri Jul 18 03:37:14 2025

solver = Normal

Maximum thread count: 8

tnom = 27

temp = 27

method = trap

.OP point found by inspection.

.step cpar=5e-10

.step cpar=1e-09

.step cpar=1.5e-09

Total elapsed time: 0.064 seconds.

Files loaded:


C:\Users\DELL\Desktop\Siemens AMS 2025\lab1\simple\_rc\_ac.cir

Measurement: peak

step	MAX(mag(V(2)))	FROM	TO
1	(-4.28633915609e-11dB, 0°)	1	100000000
2	(-1.71452601918e-10dB, 0°)	1	100000000
3	(-3.85768595401e-10dB, 0°)	1	100000000

Measurement: bw

step	mag(V(2))=PEAK/SQRT(2)
1	318360.932452
2	159171.763992
3	106165.634071



# Transient

---

## Simple RC Circuit

### \* Circuit Description

#### \* Parameters

\*\*\* add line here \*\*\*

.PARAM CPAR=500p

#### \* Signal sources

V1 1 0 PULSE(0V 1V 0 100n 100n 10u 20u)

### \* Circuit elements

R1 1 2 1k

C1 2 0 {CPAR}

### \* Initial conditions

.IC V(2)=0

### \* Analysis request

\* Run transient for 40us with 100ns step

\*\*\* add line here \*\*\*

.TRAN 100n 40u

\* Use parametric sweep for CPAR: 500p:500p:1.5n

\*\*\* add line here \*\*\*

\*.STEP PARAM CPAR 500p 1.5n 500p

.STEP PARAM CPAR LIST 500p 1n 1.5n

\* Measure rise time from 10% to 90%

.MEAS TRAN TRISE

+ TRIG when v(2) = 0.1 CROSS = 1

+ TARG WHEN V(2) = 0.9 CROSS = 1

\*\*\* add line here \*\*\*

.PRINT TRAN V(1) V(2)

.PLOT TRAN V(1) V(2)

\*\*\* add line here \*\*\*

.END



SPICE Output Log: C:\Users\DELL\Desktop\Siemens AMS 2025\lab1\simple\_rc\_tran.log


```
LTspice 24.1.9 for Windows
Circuit: C:\Users\DELL\Desktop\Siemens AMS 2025\lab1\simple_rc_tran.cir
Start Time: Fri Jul 18 03:34:56 2025
solver = Normal
Maximum thread count: 8
tnom = 27
temp = 27
method = trap
.OP point found by inspection.
.step cpar=5e-10
.step cpar=1e-09
.step cpar=1.5e-09
Total elapsed time: 0.227 seconds.
```

Files loaded:

C:\Users\DELL\Desktop\Siemens AMS 2025\lab1\simple\_rc\_tran.cir

Measurement: trise

step	trise FROM	TO
1	1.09838514095e-06	1.03673538479e-07 1.20205867943e-06
2	2.19830457284e-06	1.55908759936e-07 2.35421333277e-06
3	3.2964401991e-06	2.08544980342e-07 3.50498517945e-06



## Part 2:

1-

```
OPAMP.cir OPAMP.cir
OPAMP CIRCUIT SIMULATION
*** opamp subcircuit ***
.subckt opamp 1 2 3

* VCVS with gain 10000
Eopamp 1 0 4 0 1
Gp 0 4 2 3 0.6289 ; A0 = Gp * Rp
* redundant current sources to avoid errors
Iopen1 2 0 0A
Iopen2 3 0 0A

* Parameters
Rp 4 0 15.9k ; dominant pole resistor
Cp 4 0 10n ; dominant pole capacitor

.ends opamp
```

Given :  $A_0 = 10000$  and  $UGF = 10\text{MHz}$ .  $f_p = 1\text{KHz}$

$$R_p C_p = \frac{1}{2 \times 3.14 \times f_p} = \frac{1}{2 \times 3.14 \times 1000} = 159.23 \mu$$

Assumed  $R_p = 15.9 \text{ Kohm}$  ,  $C_p = 10\text{nF}$

$$G_p = A_0 / R_p = 0.6289$$

2-

```
*** Circuit ***
*V1 IN+ 0 DC 1V ; 1V DC input
Vsig IN+ 0 SIN(0 1 1k) ; 1V amplitude, 1kHz frequency sine wave

Rin IN- 0 1k ; Input resistor
Rf IN- OUT 9k ; Feedback resistor
XOP OUT IN+ IN- opamp ; Op-amp subcircuit


*** Transfer Function Analysis ***
*.TF V(OUT) V1

*** Transient Analysis ***
.TRAN 20u 2m 20u ; Step = 20 us, Stop time = 2 ms

*** Probes ***
.PRINT TRAN V(IN+) V(OUT)
.PLOT TRAN V(IN+) V(OUT)

*** Measurement Commands ***
.MEAS TRAN Vsig_peak MAX V(IN+)
.MEAS TRAN Vout_peak MAX V(OUT)

.END
```

 C:\Users\DELL\Desktop\Siemens AMS 2025\lab1\OPAMP.cir

```
--- Transfer Function ---

transfer_function:          9.99001      transfer
V1#input_impedance:        1e+20        impedance
output_impedance_at_v(out): 0           impedance
```

We can see that transfer function =  $\frac{V_{out}}{V_{sig}} = 9.99$  and the TF analatically =10

Very large input impedance(ideally infinity) and small output impedance(ideally 0) and that matches the predicted from the opamp analysis



3-

```
*** Circuit ***
*V1 IN+ 0 DC 1V ; 1V DC input
Vsig IN+ 0 SIN(0 1 1k) ; 1V amplitude, 1kHz frequency sine wave

Rin IN- 0 1k ; Input resistor
Rf IN- OUT 9k ; Feedback resistor
XOP OUT IN+ IN- opamp ; Op-amp subcircuit

*** Transfer Function Analysis ***
*.TF V(OUT) V1

*** Transient Analysis ***
.TRAN 20u 2m 20u ; Step = 20 us, Stop time = 2 ms

*** Probes ***
.PRINT TRAN V(IN+) V(OUT)
.PLOT TRAN V(IN+) V(OUT)

.END
```

LTspice 24.1.9 for Windows

Circuit: C:\Users\DELL\Desktop\Siemens AMS 2025\lab1\OPAMP.cir

Start Time: Mon Jul 21 19:55:05 2025

solver = Normal

Maximum thread count: 8

tnom = 27

temp = 27

method = trap

.OP point found by inspection.

Total elapsed time: 0.061 seconds.

Files loaded:



C:\Users\DELL\Desktop\Siemens AMS 2025\lab1\OPAMP.cir

vdiff\_peak: MAX(V(VDIFF))=0.00141209574511 FROM 0 TO 0.00198


vsig\_peak: MAX(V(IN+))=0.999020652352 FROM 0 TO 0.00198

vout\_peak: MAX(V(OUT))=9.98029558502 FROM 0 TO 0.00198





Right-Click to manually enter Horizontal Axis Limits



4- from the measured peaks of  $V_{sig}$  and  $V_{out}$

$$A_v = V_{out} / V_{sig} = \frac{9.9802}{0.999} = 9.99$$


Hand analysis	TF analysis	Transient analysis
$A_v = 1 + \frac{9K}{1K} = 10$	From TF log $A_v = 9.9901$	$A_v = 9.99$

We can see that the analytical way gives the most accurate results although LTspice gives good results also for both TF analysis and Transient analysis

- 5- ? At **1 kHz**, the op-amp has high gain → feedback works → differential input is **very small**.
- ? At **10 MHz**, the op-amp can no longer amplify → feedback fails → differential input is **larger**

Differential voltage for frequency 1KHz

`vdiff_peak: MAX(V(VDIFF))=0.000999087070492 FROM 0 TO 1.98e-07`



The amplitude of  $V_{diff} = 1\text{mV}$  nearly .

### Hand analysis:

At low frequency, the output is:

$$V_{out} = A_v * V_{sig} = 10 * 1 = 10 \text{ V}$$


So The differential input voltage is approximately:

$$V_{diff} = V_{out} / A_0 = 10 / 10000 = 1\text{mV}$$

6-

Differential voltage for frequency 10MHz

`vdiff_peak: MAX(V(VDIFF))=0.949425053546 FROM 0 TO 1.98e-07`  
`vsig_peak: MAX(V(IN+))=0.999966450467 FROM 0 TO 1.98e-07`  
`vout_peak: MAX(V(OUT))=1.72927891716 FROM 0 TO 1.98e-07`



The amplitude of  $V_{diff} = 1V$  nearly .

### Hand analysis:

At UGF, the op-amp's open-loop gain drops to 1 :  $A_0(f=UGF) = 1$

The feedback will not work properly and the differential voltage will be close in value to output voltage with very small gain ( nearly no gain ).

### 7- Netlist :

```

test.cir test.cir
OPAMP CIRCUIT SIMULATION
*** opamp subcircuit ***
.subckt opamp 1 2 3
* VCVS with gain 10000
Eopamp 1 0 4 0 1
Gp 0 4 2 3 0.6289 ; A0 = Gp * Rp
* redundant current sources to avoid errors
Iopen1 2 0 0A
Iopen2 3 0 0A

* Parameters
Rp 4 0 15.9k ; dominant pole resistor
Cp 4 0 10n ; dominant pole capacitor

.ends opamp

*** Circuit ***
* AC source for AC analysis
Vsig IN+ 0 AC 1
Rin IN- 0 1k ; Input resistor
Rf IN- OUT {R_feedback} ; Feedback resistor, using a parameter for sweep
XOP OUT IN+ IN- opamp ; Op-amp subcircuit


*** AC Analysis ***
.AC DEC 10 1 100Meg ; Decade sweep, 10 points per decade, from 1Hz to 100MHz

*** Parametric Sweep ***
.STEP PARAM R_feedback LIST 9k 4k

*** Probes ***
.PRINT AC V(OUT)
.PLOT AC DB(V(OUT))


```

## Waveform :




## GAIN

For the case  $R_F = 4k$  : GAIN= 14dB



For the case  $R_F = 9K$  : GAIN = 20dB




## 3dB Bandwidth

For the case  $R_F = 4k$  :  $BW = 2MHz$




For the case  $R_F = 9k$  :  $BW = 1MHz$




Right Click to manually enter Horizontal Axis Limits

## Unity Gain Frequency :

For the case  $R_F = 4k$  :  $UGF = 9.85MHz$  (nearly)



For the case  $R_F = 9k$  :  $UGF = 9.95MHz$  (nearly)





## 8- AC Analysis

There will be no clipping in the output because this is linear small-signal analysis that analyses the response with frequency , The simulator scales everything linearly and does not consider nonlinearities but we can achieve higher gain if the input amplitude increases .

### Transient Analysis

you can see clipping in transient analysis if the output tries to go beyond the op-amp's output voltage limits (supply rails) because this is a time-domain simulation and applies real large voltages on the circuit

## 9- DC Gain :

$$A_v = 1 + \frac{R_f}{R_{in}}$$

For  $R_F = 9K$  :  $A_v = 20\log( 1 + (9K/1K) ) = 20dB$

For  $R_F = 4K$  :  $A_v = 20\log( 1 + (4K/1K) ) = 14dB$

### Bandwidth:

$$BW = \frac{UGF}{A_v}$$

For  $R_F = 9K$  ,  $UGF = 10MHz$  :  $BW = 10M / 10 = 1MHz$

For  $R_F = 4K$  ,  $UGF = 10MHz$  :  $BW = 10M / 5 = 2MHz$

### Unity Gain Frequency:

Circuit was designed for  $UGF = 10MHz$  so it is for the two cases

Parameter	Hand Analysis	AC analysis
DC Gain	$R_F=4K \Rightarrow 14dB$ $R_F=9K \Rightarrow 20dB$	$R_F=4K \Rightarrow 13.975dB$ $R_F=9K \Rightarrow 19.99dB$
3dB Bandwidth	$R_F=4K \Rightarrow 2MHz$ $R_F=9K \Rightarrow 1MHz$	$R_F=4K \Rightarrow 2MHz$ $R_F=9K \Rightarrow 1MHz$
UGF	$R_F=4K \Rightarrow 10MHz$ $R_F=9K \Rightarrow 10MHz$	$R_F=4K \Rightarrow 9.8MHz$ $R_F=9K \Rightarrow 9.945MHz$

## BONUS ( Non Inverting Summing Amplifier ) :

### Netlist :

```
* === OPAMP BEHAVIORAL SUBCKT ===
.subckt opamp 1 2 3
Eopamp 1 0 4 0 1
Gp 0 4 2 3 0.6289
Iopen1 2 0 0A
Iopen2 3 0 0A
Rp 4 0 15.9k
Cp 4 0 10n
.ends opamp

* === SIGNAL SOURCES ===
V1 VIN1 0 SIN(0 1 1k)      ; 1V amplitude sine at 1kHz
V2 VIN2 0 SIN(0 0.5 1k)    ; 0.5V amplitude sine at 1kHz

* === SUMMING RESISTORS INTO NON-INVERTING INPUT ===
R1 VIN1 NPLUS 10k
R2 VIN2 NPLUS 10k

* === FEEDBACK NETWORK ON INVERTING INPUT ===
Rf NMIN OUT 9k
Rin NMIN 0 1k

* === OPAMP INSTANCE ===
XOP OUT NPLUS NMIN opamp

* === ANALYSIS ===
.TRAN 10u 10m
.PRINT TRAN V(VIN1) V(VIN2) V(OUT)


.MEAS TRAN VIN1_peak MAX V(VIN1)
.MEAS TRAN VIN2_peak MAX V(VIN2)
.MEAS TRAN VOUT_peak MAX V(OUT)
```

### Measurement :

```
vin1_peak: MAX(V(VIN1))=0.999999701977 FROM 0 TO 0.01
vin2_peak: MAX(V(VIN2))=0.499999850988 FROM 0 TO 0.01
vout_peak: MAX(V(OUT))=7.49249172211 FROM 0 TO 0.01
```

---

### Wave form :



### Components :

- 1- Two AC sources at frequency 1KHz with amplitudes 1 , 0.5 Volts
- 2- Two summing resistors equal 10Kohm
- 3- Feedback resistor equals 9Kohm
- 4- Input resistor equals 1Kohm

### Analysis :

$$A_v = 1 + \frac{9K}{1K} = 10$$

$$V_+ = \frac{V_1 + V_2}{2} \text{ (equal resistor weights)}$$

$$V_{out} = A_v * V_+ = 10 * 0.75 = 7.5 \text{ V}$$

$$\text{Voltage gain} = V_{out} / V_{sig} = \frac{7.5}{1+0.5} = 5$$

### TF Analysis :

We perform TF analysis on each source with the output and we find the results are identical and the voltage gain is near to the gain we obtained from Transient analysis

transfer_function:	4.995	transfer
V2#input_impedance:	20000	impedance
output_impedance_at_v(out):	0	impedance